
(モデル理論アプローチによるシミュレーション)

シミュレーションの高速化

：一次元細胞オートマトンの処理限界と表示限界
東洋大学　旭貴朗

付録  グラフによるデータ表示

1 はじめに

　本稿では， 例題として，1次元細胞オートマトン（Cellular Automaton）のシミュレーションモデルを，モデル理論アプローチによるモデル記述言語CASTを用いて構成し実行する．1次元細胞オートマトンのシミュレーションでは，全体システムの状態を数値が並んだリストで表し，状態遷移をリストの要素を書き換える計算とする．ここでは，オートマトン１回の状態遷移で，200個の細胞の１サイクル（１世代）の計算を行なう．この例題を通じて，開発実行環境Simcastにおけるシミュレーション高速化ための工夫やグラフ表示の制約を述べることが目的である．

　このサイト「モデル記述言語CASTによるシミュレーション」[注1]で公開されているsimcast.zip（＋extProlog）をインストールすると，モデル理論アプローチによるシミュレーション開発実行環境Simcastができあがり，バージョンが090901以降なら，本文中のモデルを実行することができる．Simcastの操作方法などは，同サイトや教科書[注2]を参照されたい．

2 例題：1次元細胞オートマトン

　直線上に並んだ複数の2状態オートマトン（正確には2状態の状態機械）を考える．各細胞の状態は0または1である．左から順番に番号をつけ，i番目のオートマトンを細胞i（cell i）と呼ぶ．

【システム要素】　各細胞の状態は，左右に隣り合う2つの細胞の状態の影響を受けて変化していく．

　　　　　[image: image2.wmf]
　　　　　　　　図表1　細胞iの状態遷移

　例えば，細胞iの近傍が図1左のようであったとき，次の時刻の細胞iの状態は1であるなどと遷移するのである．このような規則を，以下では次のように書くことにする．
　　[ [1,0,0]],1]

これは，自分が健康(0)であっても隣の細胞が病気(1)になったら自分も感染して病気(1)になる，などと適当に解釈する．自身を含めた近傍の様子は可能性として23=8パターンあるので，それぞれの場合に応じて次の時刻の状態を定めることになる．一つの細胞は複数の規則からなる状態遷移関数をもつ．以下では，次の規則を考える．

　　Rule.g=[ [[1,0,0],1], [[0,0,1],1], [[1,1,0],0], [[0,1,1],0] ];
【全体システム】　以下では200個の細胞を考え，すべての細胞の状態遷移関数は同じであるとする．一般に，状態機械の複合システムは全体としてひとつの状態機械であるから，全体をひとつの状態機械と考えることができる．すると，全体システムの状態は，値が0か1であるような200個の要素からなるリストとなる．状態遷移につれて，そのリストの中の数値が書き換えられていくのである．1回の状態遷移で200個すべての細胞の状態が変化する可能性もある．

　上記の全体システムのCAST言語によるモデルは次のように構成できる．


/*cellular02.set ( 1次元細胞オートマトン）*/

preprocess() <-> 


M.g:=200,


Rule.g:=[ [[1,0,0],1], [[0,0,1],1], [[0,1,1],0], [[1,1,0],0] ];

inputsequence()="d";

times()=1000;

initialstate()=c0 <-> 


L:=constantlist(0,M.g),


[c0]:=replaceMatrix([L],[[1,7,1],[1,140,1]]);

delta(c,a)=cc <-> 


Ls:=defList(p(z,i,[c]),member(i,genIndex(2,M.g-1))), //変更チェック


[cc]:=replaceMatrix([c],Ls);  //変更実施

p(z,i,[c]) <-> N:=project(c,["r",i-1,i+1]), member([N,v],Rule.g), z:=[1,i,v];
　　　　図表2　細胞オートマトンのモデル

　全体システムの実行手順は簡単である．まず初期状態c0を定めるために，constantlist関数で0がM.g個（200個）並んだリストLを作り，それを行列[L]に変えてから，replaceMatrix 関数で7番目と140番目だけを1に書き換えている．その結果のリストc0が初期状態である．実際の全体システムの状態遷移はdeltaで定義し，変化すべき細胞のリストアップ（変更チェック）と状態（リスト）の変更を実施する．そのため，defList関数と述語pを使って，近傍Nと変更規則Rule.gに従って書き換えるべき細胞の番号と次の状態値の組（[i,v]）をリストアップし，再びreplaceMatrix関数でリストの要素の書き換えを実行する．その結果のリストが全体システムの次状態ccである．（defListはiを動かして述語p(z,i,[c])を満足する要素zを集めてリストを作る関数である．）全体は自律的に動くシステムであるが，強制的に時間を進めるためにダミーの"d"を1000回入力している（inputsequence()="d"; times()=1000;）．

　このモデル（図表2）をモデル理論アプローチによるシミュレーション開発実行環境Simcastでコンパイルし実行した結果が図表3である．Dialogウインドウに次々とリストが表示され，状態遷移するようすが分かる．

図表3　実行結果
3　計算処理の制限

　細胞数を増やすには，各種の制限に注意する必要がある．

1）変数の制約

　Simcastにおける（正確に言えば，extPrologが処理できる）リストの長さは4,000までである．したがって，グローバル変数をM.g=4000に修正すれば4千個の細胞オートマトンを扱うことができるはずであるが，実際には関数constantlist()がエラーとなる．正常に動くことはM.g=1900まで確認した．1次元の細胞オートマトンなら通常1,900個を扱う必要はないのでこれで十分である．しかし実行速度は遅くなる．

2）ファイル保存
　遷移したすべての状態を，シミュレーション実行中にファイルに書き込むことができる．述語xwrite()を使用すると，例題では200要素のリストが1000行並んだファイルが出来上がる．容易にCSV形式に変換できるので，シミュレーション終了後に，そのデータを何らかの市販の表計算ソフトで読み込んでグラフ化することは可能である[注1]．

3）実行速度
　次にシミュレーションの速度について考える．これらのモデルでは，近傍（左右の細胞）の状態をチェックして変更すべき細胞をリストアップしてから，状態c（リスト）を書き換えていた．しかしながら，細胞数が増加すると近傍チェックの回数も増加していく．例えば，細胞数200個の場合は，（リストの両端はチェックしないので）1回の状態遷移のために198回のチェックを行うことになる．状態遷移を1,000回繰り返すなら，198,000回のチェックを行うことになる．実行時間は64秒であった．

　実は，背景で動いているextPrologはリストの書き換え速度が遅い．例題では，リストや行列の要素を（project関数で）読んだり,（replaceListやreplaceMany関数で）書き換えたりしている．そのたびごとにリストや行列を作成するから実行速度は遅くなる．細胞数が増加するとシミュレーション速度が遅くなり，多数の繰り返しには不向きである．その理由は「ひとつの変数cにすべての細胞の状態を記憶させている」からである．そこで，次節のようにモデル化の大胆な方針変更が必要となる．

4　シミュレーションの高速化

　シミュレーション実行速度を向上させるため，ここでは「ひとつのスプレッドシート（以下SS）に全細胞の状態を保存する」方法を紹介する．理由は，SimcastではSSの読み書き速度が速いからである．

　　　　　　[image: image4.wmf]
　　　　　　　　　図表4　SSを利用した高速化

　まず，ひとつの変数cにすべての細胞の状態を保存することをあきらめる．その代わりSSにすべての状態を書き込んでゆく．したがって，オートマトンの仕事は，SSを書き換えることとする（図表4）．この方針に従って，例題の細胞オートマトンを，書き直したものを図表5に掲載する．


//cellular06.set

preprocess() <-> //SSウインドウを開き，初期値を書き込む


M.g:=200,


Rule.g:=[ [[1,0,0],1], [[0,0,1],1], [[1,1,0],0],[[0,1,1],0] ],


getWp("sheet",Wp),


(Wp>0)->(Wp.g:=Wp)


otherwise (



openlocalsheet("",0,1,150,3,x1,y1,1,1,1,Wp1), Wp.g:=Wp1,



writess(Wp.g,"r", 1,1, constantlist(0,M.g)),



ssw(Wp.g,1,70,1), ssw(Wp.g,1,140,1)   );
inputsequence()="d";

times()=1000;

initialstate()=c <-> c:=[];

delta(c,a)=cc <-> 


cc:=defList(p1(z,i,[]),member(i,genIndex(2,M.g-1))),  //変更チェック


replacess(Wp.g,cc);  //SSに書き込み

p1(z,i,[]) <-> [N]:=projectss(Wp.g,1,i-1,1,i+1), member([N,v],Rule.g), z:=[1,i,v];

　図表5　スプレッドシートを「状態」の記憶場所として利用するモデル

　図表5では，実行前の準備proprocessとして3行150列表示のSSウインドウを開き，SSの第1行に初期値としてセル(1,70)とセル(1,140)に1を書き込んでいる．これが全体システムの初期状態に相当する．状態遷移の定義deltaでは，まずSSから近傍の様子Nを読み込み，規則Rule.gに従ってチェックし，変化すべき細胞をリストアップしている（変数cc）．次に変更（SSに書き込み）を実施している．

　これを実行すると，図表5の1000世代分の実行時間は22秒であった．前回の3分の1である．また，遷移する各時点の状態がスプレッドシートに書き込まれるので，シミュレーションの状況がいくぶん分かるようになった（図表6）．

　　　　　図表6　セル幅1のスプレッドシートを「状態」として利用する

　さらに高速化を望むなら，述語replacess(Wp.g, cc)を使わず，関数defListを使えばよい．
Ls:=defList(p2(z,[r,i,v],[]),member([r,i,v],cc));  //SSに書き込み
p2(z,[r,i,v],[]) <-> ssw(Wp.g,r,i,v);
この場合は，1000回で15秒であった（付録も参照のこと）．
　このモデルでは，変化すべき細胞のセル位置(1,i)と値vのリスト[1,i,v]を記憶する場所としてオートマトンの状態変数cを利用している．もしもこれをファイルに保存しておけば，初期状態からの再生も可能である．一方，細胞オートマトン全体の真の状態はSSそのものであり，initialstateやdeltaの仕事はSSを規則に基づいて書き換えることである．これは典型的な手続き型のプログラミングスタイルであり，もはやモデル理論アプローチの範囲を逸脱しているとも言える．その意味でこれを裏技というべきかも知れない．ただし，CAST言語の文法（構文）に従ってモデルを作成していることは確かである．

5　データ表示の限界

　研究や教育のためには処理データのビジュアルな表示が必要である，ここではデータ表示の限界を考察する．

1）スプレッドシート

　処理できるスプレッドシートのサイズは600行×200列までである．しかしサイズを大きくすると狭いパソコン画面からウインドウがはみ出すことになる．述語

　　openlocalsheet(Name,WinY,WinX,Y,X,Y0,X0,NotMove,Map,Len_cell,Wp)
でセル幅Len_cellを指定できるので，セル幅を1にすれば，12インチディスプレーでも40行×150列程度はリアルタイムで表示できる．

　例題の場合，細胞数が200を超えると，折り畳んで次の行に書き込む必要がある．番号iを200k＋lに変換すればよい．600×200 = 12万であるから，SSは十分な広さを持っている．
2) グラフとスプレッドシートの混在

　付録にshow系の述語でグラフ表示を行うユーザモデルを掲載した．図表5に述語showHistory()を加えたものである（ただし，高速化のためreplacessを使わずdefListを使っている）．ここでは，SSは状態の表示を担わず，単に状態の保存場所として確保しているにすぎない．その代わり，述語show1によって状態をグラフ表示することとした．1000回の状態遷移を実行するのに33秒かかった．グラフ表示には多少の負荷がかかる．
【制限】データをグラフ表示すると，グラフウインドウとスプレッドシートが混在することになる．Simcastでは混在は可能である．しかし，混在させるとSSのセル幅指定がデフォールトのセル幅11に戻るのである（改善は今後の課題である）．

　　　　　　　　　　　図表7　グラフによるデータ表示

　付録のモデルをコンパイルして実行すると，図表7のように，シミュレーション実行中に刻々と状態が変化していく様子が観察できる．下のウインドウout_0.gには200個の細胞のその時点の状態が表示されている．横軸は細胞番号i=1,2,3,...,200が，縦軸に状態0,1がグラフ化されている（関数show1による）．上のウインドウout_7.gには，各時点で状態1の細胞数が表示されている．横軸は時間で，縦軸は状態1の細胞数である．

　グラフウインドウに表示できる点の個数に制限がある．簡便なshow系のグラフ表示の述語show1やshow2では，400個の点までは正常に表示できるが，それ以上は無理である．細胞数が多いときや繰り返しの回数が多いときは，最新の200個だけを表示するなど工夫が必要である（計算そのものは細胞数1,900個まで正常に実行される）．

　一方，座標軸を非表示にしxwriteを使って描画するwopen系のグラフ表示の述語では，多数の点が打てるので問題はない．横軸をセル番号，縦軸を時刻にして下に伸びていくようにモデル化することも可能である．

　以上，計算はできるがデータ表示の制限があることを見た．したがって，200以上の細胞数のシミュレーションモデルを作成するときの方針としては，最初に細胞数M.g = 200でモデルを作成し，グラフ表示をさせながらバグのチェックなどを行い，次にグラフ表示を外し，M.gを200以上に設定し，実行することになる．その際，ファイルに「変化した状態cc」のみを保存しておき，シミュレーション実行後に市販ソフトでグラフ化することになろう．

6 おわりに

　本稿では，1次元細胞オートマトンのモデルをCAST言語を用いて構成し，モデル理論アプローチによるシミュレーション開発実行環境Simcastで実際に実行した．また，処理限界やその回避方法としてスプレッドシートを利用した高速化の工夫について紹介した．
　一応，グラフ表示しなければ多数の細胞数でもシミュレーションは可能である．しかし，もしもグラフ表示するならば，その表示限界はある．グラフ表示しつつモデルを作成し，実際のシミュレーションではグラフ表示をはずすことになろう． 

注

[1] 旭貴朗「モデル記述言語CASTによるシミュレーション」
　http://www2.toyo.ac.jp/~asahi/research/simulation/（2008.02.23）

[2] 高原康彦ほか『形式手法 モデル理論アプローチ：情報システム開発の基礎』日科技連(2007) の少なくとも第３章3.1節pp.87-103 を参照されたい．

　http://www.geocities.jp/kisoten/（2008.04.11）

付録　グラフによるデータ表示

//cellular07.set

preprocess() <-> //SSウインドウを開き，初期値を書き込む


M.g:=200,hist.g:=[],

Rule.g:=[ [[1,0,0],1], [[0,0,1],1], [[1,1,0],0],[[0,1,1],0] ],


getWp("sheet",Wp),


(Wp>0)->(Wp.g:=Wp)


otherwise (



makesheet(3,4,Wp1), Wp.g:=Wp1,



writess(Wp.g,"r", 1,1, constantlist(0,M.g)),



ssw(Wp.g,1,70,1), ssw(Wp.g,1,140,1)   );

inputsequence()="d";

times()=1000;

initialstate()=c <-> c:=[];

delta(c,a)=cc <-> 


cc:=defList(p1(z,i,[]),member(i,genIndex(2,M.g-1))),  //変更チェック


Ls:=defList(p2(z,[r,i,v],[]),member([r,i,v],cc));  //SSに書き込み

lambda(c)=b <-> b:=0, showHistory();

p1(z,i,[]) <-> [N]:=projectss(Wp.g,1,i-1,1,i+1), member([N,v],Rule.g), z:=[1,i,v];

p2(z,[r,i,v],[]) <-> ssw(Wp.g,r,i,v);

showHistory() <->


[c]:=projectss(Wp.g,1,1,1,M.g),


show1(c,"plot"),


hist1:=hist.g, x:=sum(c),


hist2:=append(hist1,[x]), len:=cardinality(hist2),


hist3:=project(hist2,["r", len-199,len]),


show2(7,hist3,"plot"),


//y:=invproject(c,1,"l"),(x <7)->(xwriteln(0,"place=",floor(y-1)),.stop),


hist.g:=hist3;

